Description

twoway bar displays numeric \((y,x)\) data as bars. twoway bar is useful for drawing bar plots of time-series data or other equally spaced data and is useful as a programming tool. For finely spaced data, also see \([G-2]\) graph twoway spike.

Also see \([G-2]\) graph bar for traditional bar charts and \([G-2]\) graph twoway histogram for histograms.

Quick start

Bar graph, with bars extending from 0

twoway bar y x

A horizontal bar graph

twoway bar y x, horizontal

Bar graph with bars 0.8 times the default width

twoway bar y x, barwidth(.8)

Bars that extend from 20 rather than 0

twoway bar y x, base(20)

Add the title “My Title” to the graph

twoway bar y x, title("My Title")

As above, but add a title “X variable” to the \(x\) axis

twoway bar y x, title("My Title") xtitle("X variable")

As above, but with separate graph areas for each value of \(\text{catvar}\)

twoway bar y x, xtitle("X variable") by(catvar, title("My Title"))

Menu

Graphics > Twoway graph (scatter, line, etc.)
Syntax

```
twoway bar  yvar  xvar  [if]  [in]  [,  options]
```

### Options

<table>
<thead>
<tr>
<th>Option</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>vertical</code></td>
<td>vertical bar plot; the default</td>
</tr>
<tr>
<td><code>horizontal</code></td>
<td>horizontal bar plot</td>
</tr>
<tr>
<td><code>base(#)</code></td>
<td>value to drop to; default is 0</td>
</tr>
<tr>
<td><code>barwidth(#)</code></td>
<td>width of bar in <code>xvar</code> units</td>
</tr>
<tr>
<td><code>barlook_options</code></td>
<td>change look of bars</td>
</tr>
<tr>
<td><code>axis_choice_options</code></td>
<td>associate plot with alternative axis</td>
</tr>
<tr>
<td><code>twoway_options</code></td>
<td>titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.</td>
</tr>
</tbody>
</table>

Options `base()` and `barwidth()` are rightmost, and `vertical` and `horizontal` are unique; see [G-4] concept: repeated options.

### Options

- `vertical` and `horizontal` specify either a vertical or a horizontal bar plot. `vertical` is the default. If `horizontal` is specified, the values recorded in `yvar` are treated as `x` values, and the values recorded in `xvar` are treated as `y` values. That is, to make horizontal plots, do not switch the order of the two variables specified.

  In the `vertical` case, bars are drawn at the specified `xvar` values and extend up or down from 0 according to the corresponding `yvar` values. If 0 is not in the range of the `y` axis, bars extend up or down to the `x` axis.

  In the `horizontal` case, bars are drawn at the specified `xvar` values and extend left or right from 0 according to the corresponding `yvar` values. If 0 is not in the range of the `x` axis, bars extend left or right to the `y` axis.

- `base(#)` specifies the value from which the bar should extend. The default is `base(0)`, and in the above description of options `vertical` and `horizontal`, this default was assumed.

- `barwidth(#)` specifies the width of the bar in `xvar` units. The default is `width(1)`. When a bar is plotted, it is centered at `x`, so half the width extends below `x` and half above.

- `barlook_options` set the look of the bars. The most important of these options is `color(colorstyle)`, which specifies the color of the bars; see [G-4] `colorstyle` for a list of color choices. See [G-3] `barlook_options` for information on the other `barlook_options`.

- `axis_choice_options` associate the plot with a particular `y` or `x` axis on the graph; see [G-3] `axis_choice_options`.

- `twoway_options` are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] `twoway_options`. 
Typical use

We have daily data recording the values for the S&P 500 in 2001:

. use http://www.stata-press.com/data/r14/sp500 (S&P 500)
. list date close change in 1/5

<table>
<thead>
<tr>
<th>date</th>
<th>close</th>
<th>change</th>
</tr>
</thead>
<tbody>
<tr>
<td>02jan2001</td>
<td>1283.27</td>
<td>.</td>
</tr>
<tr>
<td>03jan2001</td>
<td>1347.56</td>
<td>64.29004</td>
</tr>
<tr>
<td>04jan2001</td>
<td>1333.34</td>
<td>-14.22009</td>
</tr>
<tr>
<td>05jan2001</td>
<td>1298.35</td>
<td>-34.98999</td>
</tr>
<tr>
<td>08jan2001</td>
<td>1295.86</td>
<td>-2.48999</td>
</tr>
</tbody>
</table>

We will use the first 57 observations from these data:

. twoway bar change date in 1/57
We get a different visual effect if we reduce the width of the bars from 1 day to .6 days:

```
   . twoway bar change date in 1/57, barw(.6)
```

![Graph showing closing price change over time with bar width of 0.6 days]

**Advanced use: Overlaying**

The useful thing about `twoway bar` is that it can be combined with other `twoway` plottypes (see [G-2] `graph twoway`):

```
   . twoway line close date || bar change date || in 1/52
```

![Graph showing overlay of line and bar plots]

---

A detailed explanation of the code and its output can be found in the documentation for `graph twoway`. The `twoway bar` command allows for customizing the appearance of the bars, and the `twoway line` command can be used to add line plots for additional data visualization.
We can improve this graph by typing

```
. twoway
    line close date, yaxis(1)
    ||
    bar change date, yaxis(2)
    ||
in 1/52,
yasca(axis(1) r(1000 1400)) ylab(1200(50)1400, axis(1))
yasca(axis(2) r(-50 300)) ylab(-50 0 50, axis(2))
ytick(-50(25)50, axis(2) grid)
legend(off)
xtitle("Date")
title("S&P 500")
subtitle("January - March 2001")
note("Source: Yahoo!Finance and Commodity Systems, Inc.")
yline(1150, axis(1) lstyle(foreground))
```

Notice the use of

```
yline(1150, axis(1) lstyle(foreground))
```

The 1150 put the horizontal line at $y = 1150$; `axis(1)` stated that $y$ should be interpreted according to the left $y$ axis; and `lstyle(foreground)` specified that the line be drawn in the foreground style.
Advanced use: Population pyramid

We have the following aggregate data from the U.S. 2000 Census recording total population by age and sex. From this, we produce a population pyramid:

```
. list agegrp maletotal femtotal

<table>
<thead>
<tr>
<th>agegrp</th>
<th>maletotal</th>
<th>femtotal</th>
</tr>
</thead>
<tbody>
<tr>
<td>1. Under 5</td>
<td>9,810,733</td>
<td>9,365,065</td>
</tr>
<tr>
<td>2. 5 to 9</td>
<td>10,523,277</td>
<td>10,026,228</td>
</tr>
<tr>
<td>3. 10 to 14</td>
<td>10,520,197</td>
<td>10,007,875</td>
</tr>
<tr>
<td>4. 15 to 19</td>
<td>10,391,004</td>
<td>9,828,886</td>
</tr>
<tr>
<td>5. 20 to 24</td>
<td>9,687,814</td>
<td>9,276,187</td>
</tr>
<tr>
<td>6. 25 to 29</td>
<td>9,798,760</td>
<td>9,582,576</td>
</tr>
<tr>
<td>7. 30 to 34</td>
<td>10,321,769</td>
<td>10,188,619</td>
</tr>
<tr>
<td>8. 35 to 39</td>
<td>11,318,696</td>
<td>11,387,968</td>
</tr>
<tr>
<td>9. 40 to 44</td>
<td>11,129,102</td>
<td>11,312,761</td>
</tr>
<tr>
<td>10. 45 to 49</td>
<td>9,889,506</td>
<td>10,202,898</td>
</tr>
<tr>
<td>11. 50 to 54</td>
<td>8,607,724</td>
<td>8,977,824</td>
</tr>
<tr>
<td>12. 55 to 59</td>
<td>6,508,729</td>
<td>6,960,508</td>
</tr>
<tr>
<td>13. 60 to 64</td>
<td>5,136,627</td>
<td>5,668,820</td>
</tr>
<tr>
<td>14. 65 to 69</td>
<td>4,400,362</td>
<td>5,133,183</td>
</tr>
<tr>
<td>15. 70 to 74</td>
<td>3,902,912</td>
<td>4,954,529</td>
</tr>
<tr>
<td>16. 75 to 79</td>
<td>3,044,456</td>
<td>4,371,357</td>
</tr>
<tr>
<td>17. 80 to 84</td>
<td>1,834,497</td>
<td>3,110,470</td>
</tr>
</tbody>
</table>
```

```
. replace maletotal = -maletotal/1e+6
(17 real changes made)
. replace femtotal = femtotal/1e+6
(17 real changes made)
. twoway
   bar maletotal agegrp, horizontal xvarlab(Males)
   ||
   bar femtotal agegrp, horizontal xvarlab(Females)
   ||
   , ylabel(1(1)17, angle(horizontal) valuelabel labsize(*.8))
   xtitle("Population in millions") ytitle(""")
   xlabel(-10 "10" -7.5 "7.5" -5 "5" -2.5 "2.5" 2.5 5 7.5 10)
   legend(label(1 Males) label(2 Females))
   title("US Male and Female Population by Age")
   subtitle("Year 2000")
   note("Source: U.S. Census Bureau, Census 2000, Tables 1, 2 and 3", span)
```
At its heart, the above graph is simple: we turned the bars sideways and changed the male total to be negative. Our first attempt at the above was simply

```
. replace maletotal = -maletotal
   (17 real changes made)
. twoway bar maletotal agegrp, horizontal ||
      bar femtotal agegrp, horizontal
```

From there, we divided the population totals by 1 million and added options.

```
xlabel(-10 "10" -7.5 "7.5" -5 "5" -2.5 "2.5" 2.5 5 7.5 10) was a clever way to disguise that the bars for males extended in the negative direction. We said to label the values −10, −7.5, −5, −2.5, 2.5, 5, 7.5, and 10, but then we substituted text for the negative numbers to make it appear that they were positive. See [G-3] axis_label_options.

Using the span suboption to note() aligned the text on the left side of the graph rather than on the plot region. See [G-3] textbox_options.
```
For another rendition of the pyramid, we tried

```
. replace maletotal = -maletotal/1e+6
    (17 real changes made)
. replace femtotal = femtotal/1e+6
    (17 real changes made)
. generate zero = 0
. twoway
    bar maletotal agegrp, horizontal xvarlab(Males)
    ||
    bar femtotal agegrp, horizontal xvarlab(Females)
    ||
    sc agegrp zero , mlabel(agegrp) mlabcolor(black) msymbol(i)
    ||
    , xtitle("Population in millions") ytitle(""")
plotregion(style(none))                                  (note 1)
y sca(noline) ylabel(none)                                (note 2)
x sca(noline titlegap(-3.5))                             (note 3)
x label(-12 "12" -10 "10" -8 "8" -6 "6" -4 "4" 4(2)12 ,
        length(0) grid gmin gmax)
legend(label(1 Males) label(2 Females)) legend(order(1 2))
title("US Male and Female Population by Age, 2000")
note("Source: U.S. Census Bureau, Census 2000, Tables 1, 2 and 3")
```

In the above rendition, we moved the labels from the $x$ axis to inside the bars by overlaying a scatter on top of the bars. The points of the scatter we plotted at $y = \text{agegrp}$ and $x = 0$, and rather than showing the markers, we displayed marker labels containing the desired labelings. See [G-3] marker_label_options.

We also played the following tricks:

1. `plotregion(style(none))` suppressed outlining the plot region; see [G-3] region_options.
2. `ysca(noline)` suppressed drawing the $y$ axis—see [G-3] axis_scale_options—and `ylabel(none)` suppressed labeling it—see [G-3] axis_label_options.
3. `xsca(noline titlegap(-3.5))` suppressed drawing the $x$ axis and moved the $x$-axis title up to be in between its labels; see [G-3] axis_scale_options.
Cautions

You must extend the scale of the axis, if that is necessary. Consider using `twoway bar` to produce a histogram (ignoring the better alternative of using `twoway histogram`; see [G-2] `graph twoway histogram`). Assume that you have already aggregated data of the form

<table>
<thead>
<tr>
<th>x</th>
<th>frequency</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>400</td>
</tr>
<tr>
<td>2</td>
<td>800</td>
</tr>
<tr>
<td>3</td>
<td>3,000</td>
</tr>
<tr>
<td>4</td>
<td>1,800</td>
</tr>
<tr>
<td>5</td>
<td>1,100</td>
</tr>
</tbody>
</table>

which you enter into Stata to make variables `x` and `frequency`. You type

```
. twoway bar frequency x
```

to make a histogram-style bar chart. The `y` axis will be scaled to go between 400 and 3,000 (labeled at 500, 1,000, ..., 3,000), and the shortest bar will have zero height. You need to type

```
. twoway bar frequency x, yscale(r(0))
```

Reference


Also see

[G-2] `graph twoway scatter` — Twoway scatterplots

[G-2] `graph twoway dot` — Twoway dot plots

[G-2] `graph twoway dropline` — Twoway dropped-line plots

[G-2] `graph twoway histogram` — Histogram plots

[G-2] `graph twoway spike` — Twoway spike plots

[G-2] `graph bar` — Bar charts